3.2 Операции с полиномами

В математических преобразованиях часто используются полиномы. Под полиномом Maple понимает сумму или разность выражений с неотрицательными степенями. В пакете представлен широкий спектр команд для работы с полиномами. Ниже приведены основные из них.

команда	описание
lcoeff(pol,opt)	Определение старшего коэффициента полинома **pol**
coeff(pol,x,n)	Определение коэффициента полинома **pol** при **n**-ой степени переменной **x**
coeffs(pol,x)	Определение всех коэффициентов полинома **pol** при переменной **x**
convert(pol, sqrfree,x)	Разложение полинома **pol** на квадратные трехчлены по переменной **x**
degree(pol,x)	Определение степени полинома **pol** при переменной **x**
discrim(pol,x)	Вычисление дискриминанта полинома **pol** по переменной **x**
gcd(pol1,pol2)	Вычисление наибольшего общего делителя двух полиномов **pol1** и **pol2**
psqrt(pol)	Вычисление квадратного корня из полинома **pol**
quo(pol1,pol2,x)	Вычисление частного от деления двух полиномов **pol1** и **pol2** по переменной **x**
randpoly(x)	Создание случайного полинома переменной **x**
rem(pol1,pol2,x)	Вычисление остатка от деления двух полиномов **pol1** и **pol2** по переменной **x**

Определим полином pol:

```
> pol:=expand((5*y*x^2+x+1)*(x^3-
x)+(2*y*x^2+6));
```

$$pol := 5\,y\,x^5 - 5\,y\,x^3 + x^4 - x^2 + x^3 - x + 2\,y\,x^2 + 6$$

Выполним несколько арифмитических операций. С помощью команды quo разделим один полином на другой, определив при этом целую часть от деления:

```
> quo(pol,x^3-x,x);
```

$$5\,y\,x^2 + x + 1$$

С помощью команды rem найдем остаток от этого деления, при этом целая часть запишется в переменную `k`:

```
> os:=rem(pol,x^3-x,x,'k');k;
```

$$os := 2\,y\,x^2 + 6$$

$$5\,y\,x^2 + x + 1$$

Для нахождения наибольшего общего делителя двух полиномов используется команда gcd.

```
> gcd(os,y*x^2+3);
```

$$y\,x^2 + 3$$

Сгруппируем ранее определенный нами полином по переменной **x** с помощью команды collect и подставим вместо переменной **y** число 2.

```
> collect(pol,x); pol1:=subs(y=2,pol);
```

$$5\,y\,x^5 + x^4 + (1 - 5\,y)\,x^3 + (-1 + 2\,y)\,x^2 - x + 6$$

$$pol1 := 10\,x^5 - 9\,x^3 + x^4 + 3\,x^2 - x + 6$$

Определим коэффициенты полинома pol1и далее вычислим для него дискриминант и степень, используя соответственно команды coeffs, discrim, degree.

```
> coeffs(pol1,x); discrim(pol1,x);
degree(pol1,x);
```

10, -1, 1, -9, 3, 6

63388974740

5

Командой factor можно разложить полином на множители. Например:

```
> factor(3*x^2+9*x-12);
```

$$3\,(x + 4)\,(x - 1)$$

3.3 Ограничения на переменные

В Maple можно наложить различные ограничения на переменные, при этом программа будет предупреждать вас о наложенных ограничениях в различной форме. Режимы предупреждения подробно описаны в главе “Интерфейс Maple V”.

Пусть переменная “a” будет больше нуля:

```
> assume( a>0 );
> signum( a );
```

$$1$$

```
> Re(a+1);
```

$$a\sim + 1$$

После переменной “a” мы видим символ “~” , означающий, что на переменную наложены ограничения.

Предупреждающая надпись может быть и такого рода (если установить пункт меню *Options/Assumed Variables/Phrase)*:

```
> assume(n, integer); cos(n*Pi);
```

$$(-1)^n$$

with assumptions on n

3.4 Примеры из курса математического анализа

Основу курса математического анализа составляют такие понятия как пределы, производные, первообразные функций, интегралы разных видов, ряды и дифференциальные уравнения.

Кто знаком с основами математического анализа, тому наверняка известны десятки правил нахождения пределов, взятия интегралов, нахождения производных и т.д. и т.п. Если добавить к этому, что для нахождения большинства интегралов нужно еще помнить таблицу интегралов, то получается просто огромный объем информации. И, если некоторое время не тренировать себя в решениях таких задач, то в итоге многое забывается и для нахождения, например, интеграла посложнее уже требуется искать его в справочнике.

Но ведь взятие интегралов и нахождение пределов - это не самоцель вычислений. Всякая задача должна решать некоторую практическую проблему, а промежуточные вычисления - лишь промежуточный этап на пути к ответу.

С помощью Maple можно сэкономить массу времени и избежать многих ошибок при вычислениях.

Рассмотрим на примерах команду limit, которая позволяет находить пределы функций.

```
> restart;
> f(x):=(x^3-3*x^2+2*x-5)/(x^2+2);
```

$$f(x) := \frac{x^3 - 3x^2 + 2x - 5}{x^2 + 2}$$

```
> Limit(f(x),x=-1)=limit(f(x),x=-1);
```

$$\lim_{x \to (-1)} \frac{x^3 - 3x^2 + 2x - 5}{x^2 + 2} = \frac{-11}{3}$$

```
> r:=5*sin((3*x)/(x-Pi));
```

$$r := 5\sin\left(3\frac{x}{x - \pi}\right)$$

```
> limit(r,x=Pi);
```

$$-5 .. 5$$

Последний результат означает, что предел не найден, но значения функции в окрестности указанной точки принимают значения в диапазоне, который вычислила функция limit. Проверить это можно с помощью графика функции.

```
> plot(r,x=Pi-0.3..Pi+0.3);
```



Иногда Maple не может найти предел, например:

```
> limit(tan(x), x=infinity);
```

$$undefined$$

Можно также находить односторонние пределы. Для этого достаточно указать ключевое слово left или right.

```
> g(x):=1/x;
```

$$g(x) := \frac{1}{x}$$

```
> Limit(g(x),x=0,left)=limit(g(x),x=0,left);
```

$$\lim_{x \to 0-} \frac{1}{x} = -\infty$$

Проверить правильность нахождения предела можно с помощью графика.

```
> plot(g(x),x=-1..1,view=[-1..1,-50..50],
```

```
discont=true,color=blue,labels=[`x`,`g(x)`]);
```



Можно также находить пределы для функций нескольких аргументов:

```
> limit(x+1/y, {x=0,y=infinity});
```

$$0$$

Для Maple не составит труда найти предел для функции с неизвестными параметрами:

```
> limit(a*x, x=infinity);
```

$$\mathrm{signum}(a)\,\infty$$

Чтобы продифференцировать функцию, достаточно воспользоваться командой diff.

```
> restart:
> f(x):=ln(sqrt(exp(3*x)/(1+exp(3*x))));
```

$$\mathrm{f}(x) := \ln\left(\sqrt{\frac{e^{(3\,x)}}{1+e^{(3\,x)}}}\right)$$

```
> simplify(diff(f(x),x));
```

$$\frac{3}{2}\,\frac{1}{1+e^{(3\,x)}}$$

Можно взять частные производные от функции многих переменных:

```
> diff(f(x,y),x,y);
```

$$\frac{\partial^2}{\partial y\,\partial x}\,\mathrm{f}(x,y)$$

При помощи оператора формирования последовательности ($) можно брать производные высоких порядков.

```
> Diff(sin(x),x$3)=diff(sin(x),x$3);
```

$$\frac{\partial^3}{\partial x^3}\sin(x) = -\cos(x)$$

Взять интеграл от какой-либо функции можно при помощи оператора int.

```
> restart:
```

Неопределенный интеграл:

```
> Int((3*x^2+8)/(x^3+4*x^2+4*x), x)=
int((3*x^2+8)/(x^3+4*x^2+4*x),x);
```

$$\int \frac{3\,x^2+8}{x^3+4\,x^2+4\,x}\,dx = 2\ln(x) + \frac{10}{x+2} + \ln(x+2)$$

Определенный интеграл:

```
> Int(sin(phi)^3*sqrt(cos(phi)),phi=0..Pi/
2)=int(sin(phi)^3*sqrt(cos(phi)),phi=0..Pi/2);
```

$$\int_0^{\frac{1}{2}\pi} \sin(\phi)^3\sqrt{\cos(\phi)}\,d\phi = \frac{8}{21}$$

Несобственные интегралы (первого и второго рода):

```
> Int(1/(x^2+2*x+2),x=-infinity..infinity)=
```

```
int(1/(x^2+2*x+2),x=-infinity..infinity);
```

$$\int_{-\infty}^{\infty} \frac{1}{x^2 + 2x + 2}\, dx = \pi$$

```
> Int(1/(x-1)^2,x=0..2)=int(1/(x-1)^2,x=0..2);;
```

$$\int_{0}^{2} \frac{1}{(x-1)^2}\, dx = \infty$$

В тех случаях, когда интеграл не может быть вычислен в численной форме, поможет функция evalf.

```
> ww:=int( exp(-x^3), x = 0..1 );
```

$$ww := \int_{0}^{1} \mathrm{e}^{(-x^3)}\, dx$$

```
> evalf(ww,5);
```

$$.80751$$

В Maple можно при помощи команды sum находить предел сходимости ряда.

```
> Sum(7^(3*n)/(2*n-5)!,n=3..infinity)=
evalf(sum(7^(3*n)/(2*n-5)!,n=3..infinity));
```

$$\sum_{n=3}^{\infty} \frac{7^{(3n)}}{(2n-5)!} = .1203515449\ 10^{15}$$

Полезной может оказаться и функция product:

```
> product( a[k], k=0..n );
```

$$\prod_{k=0}^{n} a_k$$

Достаточно просто производить различные операции над кусочно-аналитическими функциями. Их можно интегрировать, дифференцировать и даже использовать в дифференциальных уравнениях.

```
> p:=piecewise(x<0, -1, x>1, 2*x, x^2);
```

$$p := \begin{cases} -1 & x < 0 \\ 2\,x & 1 < x \\ x^2 & otherwise \end{cases}$$

Построим график вышеописанной функции:

```
> plot(p,x=-1..2);
```



Интегрировать и дифференцировать кусочную функцию можно также, как и обычные функции:

```
> int(p,x);
```

$$\begin{cases} -x & x \le 0 \\ \frac{1}{3}x^3 & x \le 1 \\ x^2 - \frac{2}{3} & 1 < x \end{cases}$$

А теперь попробуем решить дифференциальное уравнение, в которое входит функция p(x).

```
> dsolve( {diff(y(x),x)+p*y(x),y(0)=-2}, y(x));
```

$$y(x)=\begin{cases} -2\,e^{x} & x \le 0 \\ -2\,e^{\left(-\frac{1}{3}x^{3}\right)} & x \le 1 \\ -2\,e^{\left(-x^{2}+\frac{2}{3}\right)} & 1 < x \end{cases}$$

Maple не ограничивается вышеописанными возможностями. Т.к. система содержит операторы для базовых вычислений, то почти все алгоритмы, которые не реализованы стандартными функциями можно реализовать посредством написания собственной программы.